

Оглавление

Введение	5
Глава 1. ЗАЧЕМ НУЖНА ФИНАНСОВАЯ ЦЕЛЬ	7
Глава 2. ЛИЧНЫЙ ФИНАНСОВЫЙ КАЛЬКУЛЯТОР	10
Глава 3. ОБЛИГАЦИОННЫЙ ПОРТФЕЛЬ	52
Глава 4. ПОРТФЕЛЬ АКЦИЙ	79
Глава 5. ПОРТФЕЛЬ ФОНДОВ	119
Глава 6. ОЦЕНКА ЭФФЕКТИВНОСТИ ПОРТФЕЛЯ	124
Глава 7. ОБЪЕДИНЕННЫЙ ПОРТФЕЛЬ	139
Глава 8. КАК ЧУВСТВУЕТ СЕБЯ КОМПАНИЯ?	142
Глава 9. СКОЛЬКО СТОИТ КОМПАНИЯ?	190
Заключение	204
Предметный указатель	205

Введение

Эта книга родилась из моих курсов, бесед с инвесторами и бесконечных вопросов клиентов, на которые я отвечала в течение последних несколько лет. Наконец-то я решилась свести все воедино и попытаться изложить основные принципы составления инвестиционного портфеля для частного инвестора просто и доступно.

Я знаю, что рынок переполнен литературой о том, как быстро и без риска озолотиться на фондовом рынке, и подозреваю, что мой скромный труд впечатлит далеко не всех. К сожалению, у меня нет волшебных методик и 100% гарантий успеха. Я не знаю, где на рынке спрятана кнопка «бабло», и не буду обещать, что у вас все получится. Но я хочу рассказать вам об инструментах, которые призваны помочь принимать более рациональные и обоснованные решения. Разумеется, эти решения могут оказаться ошибочными, однако всегда лучше понимать, почему ты принимаешь то или иное решение, чтобы в случае неудачи сделать соответствующие выводы.

Я не верю в интуитивный трейдинг, не верю в хайп, не верю в то, что рынок может сделать кого-то богатым. Обычно эта опция отсутствует в базовой комплектации, однако рынок может помочь сохранить и преумножить ваши накопления, если вы будете дисциплинированы, последовательны и терпеливы.

В этой книге будет много вычислений и расчетов. Кому-то покажется это скучным, но хочу напомнить, что рынок — это не только искусство, но и наука. Многие инструменты и процессы на рынке описаны математически. Эти модели работают и, судя по тому, насколько часто их используют профессионалы в своей работе, работают неплохо.

Я попытаюсь не просто описать сложные вещи простыми словами, но и продемонстрировать готовые модели, помогающие даже неспециалисту производить необходимые в процессе инвестирования расчеты.

Для того чтобы максимально облегчить участь читателей, большинство расчетов будут произведены с использованием Excel. Эта многофункциональная программа вполне по силам любому, кто проявит интерес и терпение.

На этом со вступлением все. Я хотела бы создать максимально утилитарную книгу. Поэтому меньше пафосных слов и больше полезной информации!

Глава 1

Зачем нужна финансовая цель

Думаю, словосочетание «финансовая цель» звучит в последнее время так часто, что у бедных инвесторов от него сводит челюсть. И в самом деле, какое отношение эта скучная бессмыслица имеет к такому веселому занятию, как инвестирование! Все постоянно призывают эту самую цель иметь, но никто не может толком объяснить, зачем это нужно.

Я сделаю попытку объяснить важность финансовой цели для правильного формирования вашего портфеля. Надеюсь, у меня это получится убедительнее, чем у некоторых моих коллег.

Итак, почему важно не просто поставить финансовую цель, но и обязательно ее оцифровать? Один умный человек, к сожалению, не помню кто, сказал: «Если человек на вопрос, сколько ему нужно денег, отвечает «много», то деньги ему не нужны. Потому что когда деньги действительно нужны, то человек всегда точно знает сколько». Полностью согласна. Если нет цели, то как понять, куда идти? Цель должна быть понятна и иметь реальное денежное выражение. Неважно, в какой валюте она будет выражена — в рублях, долларах или в мозамбикских метикалах — это должна быть конкретная сумма, которой мы хотим достичь при помощи нашей инвестиционной стратегии.

Если вы говорите «Хочу машину!», — это мечта. Если вы говорите «Хочу белый мерседес 2020 года выпуска за 2 750 000 рублей», — это ЦЕЛЬ.

Основные этапы формирования портфеля выглядят примерно так.

- 1. Оцифровываем финансовые цели.** Определяемся с инвестиционным горизонтом, начальными суммами, планируемыми пополнениями или изъятиями.

2. **Рассчитываем желаемую ставку**, то есть ту доходность, которая при заданных условиях позволит достичь поставленной финансовой цели в определенные сроки.
3. **Проводим корректировку**, если есть такая необходимость.
4. **Определяем допустимый уровень риска**. Этап сугубо индивидуальный и во многом зависит от личных предпочтений инвестора.
5. **Определяемся со структурой портфеля** (разбивка по классам активов). Это предварительная структура, и мы к ней будем возвращаться для возможных корректировок.
6. **Формируем облигационный портфель**. Его характеристики прочно связаны с вашими финансовыми целями и восприятием риска.
7. **Формируем портфель акций**, основываясь как на фундаментальных показателях, так и на историческом соотношении риска и доходности по каждому активу. Наша задача — собрать максимально диверсифицированный портфель с заданным соотношением риска и доходности.
8. **Объединяем облигационный портфель и портфель акций**. Считаем соотношение риска и доходности общего портфеля. Если есть необходимость, корректируем структуру портфеля.

Это очень краткое и схематичное описание основных этапов формирования портфеля. Оно же является и скелетом этой книги, описанием ее структуры.

Как вы могли заметить, все начинается не с открытия брокерского счета и хаотичных покупок всего подряд!

Все начинается с *финансовой цели*. Именно эта информация дает нам два важнейших параметра: сумму и срок. Ту сумму, которой мы хотели бы обладать через определенный промежуток времени.

И тут перед инвестором встает первая проблема: «Как рассчитать свою финансовую цель?» Представим, у Пети есть 1 000 000 рублей, и он хотел бы его инвестировать. Его цель — первый взнос за квартиру через три года. Ему будет нужно 3 000 000 рублей. Вроде бы и сумма есть ко-



Основные этапы формирования портфеля




Иллюстрация 1

нечная, и срок, а что дальше делать не очень понятно. Именно поэтому в главе 2 мы будем учиться считать, используя возможности Excel. И я обещаю, что после этой главы никакие финансовые калькуляторы вам больше искать не придется!

Глава 2

Личный финансовый калькулятор

Немного об Excel и расчетах

Прежде чем вы закроете книгу, испугавшись расчетов, прочитайте до конца эту страничку. К сожалению, в большинстве своем люди не очень любят считать. Обычно это не является большой проблемой, человек отлично умеет приспосабливаться. Обычно... Но не тогда, когда дело касается денег. Я общалась с огромным количеством инвесторов, которые не только не могут назвать все свои активы, не знают, на какую сумму у них инвестиций, но иногда даже не могут определить, в плюсе или минусе их портфели. На что рассчитывают эти люди? На то, что «рынок всегда растет»? Или что денюжки как-то сами собой заработаются?

Инвестирование — это такая же работа, как и многие другие, только гораздо более конкурентная. Помимо вас за деньги бьются еще «500 миллионов инвесторов», которых не стоит списывать со счетов. И если вы изначально настроены на легкую прогулку, стоит доверить свои деньги профессионалам. Это именно тот случай, когда заплатить за услугу будет дешевле.

Но если вы настроены решительно и готовы взять контроль над своими инвестициями в собственные руки, то умение считать станет вашим конкурентным преимуществом. Ведь как сказал один из героев культового фильма о рынках «Марджин колл»: «Чтобы заработать на рынке, есть только три пути: быть быстрее всех, быть умнее всех или обмануть». Быть быстрее всех для рядового инвестора крайне сложно. Обманывать я вас точно призывать не буду. А вот использовать то, что наука о финансах и инвестициях может предложить любому, кто проявит интерес, очень советую.

Итак, начнем мы с оцифровывания финансовых целей. И, как я говорила ранее, для этого мы будем использовать электронные таблицы Excel. Моя задача — максимально полезно и емко рассказать вам о тех возможностях программы, которые могут пригодиться для расчетов. И, конечно, закрепить новые знания разбором примеров.



Функция БС (FV) (будущая стоимость)

В Excel есть пять основных функций, которые помогут оцифровать наши финансовые цели. Все они относятся к аннуитетным платежам. Что это такое и чем аннуитетные платежи отличаются от всех прочих? С аннуитетными платежами сталкивались все, кто когда-либо размещал деньги на депозит или брал кредит. Это платеж **равными суммами** через **равные промежутки времени**. Наверное, пока звучит сложно, но примеры все прояснят.

А сейчас поговорим об основных параметрах, которые нам потребуются для расчетов.

Название функции	Название функции (англ)	Описание функции
ПС	PV	Современная (приведенная) стоимость будущих денежных потоков
БС	FV	Будущая стоимость денежного потока
ПЛТ	PMT	Регулярный поток платежей
СТАВКА	RATE	Процентная ставка для наращивания/дисконтирования потока платежей
КПЕР	NPER	Количество периодов

Иллюстрация 2

ПС (PV) — настоящая стоимость денежного потока. Это ваша первоначальная сумма, которую вы намерены инвестировать. Она может быть любой, в том числе и равной нулю. Принял человек решение начать откладывать, а накоплений нет. Очень жизненная, между прочим, ситуация.

БС (FV) — это будущая стоимость ваших инвестиций.

ПЛТ (PMT) — регулярный поток платежей. Это для того случая, когда вы не просто вложили первоначальную сумму в самом начале и успокоились, а продолжаете регулярно доносить на счет денежные средства. Например, ежемесячно по 10 000 рублей. Или по 30 000 рублей ежеквартально. Поскольку пока мы рассчитываем именно аннуитетные платежи, то предполагаем, что эти суммы не меняются. Они могут быть и равны нулю. Такое тоже возможно.

СТАВКА (RATE) — ставка, под которую инвестируются средства. Это самый скользкий момент во многих случаях, поэтому о ставке мы будем говорить много и часто.

КПЕР (NPER) — количество периодов инвестирования. С этим показателем есть нюансы. Excel не умеет распознавать дни, недели или месяцы, если ему в этом не «помочь», — он считает периодами. Каждый период происходит капитализация накопленных процентов, и об этом важно помнить.

Для дальнейших расчетов примем:

- n — количество лет, предположительный срок инвестирования;
- m — количество начислений в году. В аннуитетных платежах используют сложный процент при расчетах. Каждый раз, когда вы вносите деньги, происходит капитализация (то есть накопленные проценты прибавляются к основной сумме). И в следующем периоде проценты начисляются уже на новую основную сумму. Если вы не довносите деньги на счет, то капитализация будет раз в год, а $m = 1$.

А теперь простая задача.

10 000 рублей помещены в банк на депозит под 6% годовых на 3 года. Сложные проценты начисляются раз в год в конце каждого периода. Как найти величину депозита в конце срока?

$$FV = PV \cdot (1 + r)^n$$

Вот такую формулу обычно используют для расчета по этой задачке, но мы пойдем другим путем. Для расчета будущей стоимости в EXCEL используют функцию **БС** — **будущая стоимость** (или в англоязычной версии FV).

В этом примере деньги не довносят. Значит, ПЛТ (PMT) (платеж) = 0. Начисление процентов раз в год: $m = 1$. Начальная сумма на счету 10 000 рублей ПС (FV) (приведенная стоимость) = 10 000. Ставка по депозиту равна 6% годовых **СТАВКА (RATE)** = 6%.



	A	B	C	D
1				
2		Приведенная стоимость (ПС)	10 000.00 Р	
3		Будущая стоимость (БС)	11 910.16 Р	=БС(C5/C8;C6;-C4;-C2;0)
4		Платежи (ПЛТ)	0.00 Р	
5		Ставка (СТАВКА)	6.00%	
6		Количество периодов (КПЕР)	3	
7		Количество лет (п)	3	
8		Количество начислений в году (m)	1	
9				

Иллюстрация 3

Небольшое замечание: чтобы EXCEL считал корректно, все деньги, которые мы «выпускаем из рук», необходимо вводить со знаком **минус**. В данном случае, размещая деньги на депозит, мы их «выпускаем из рук». Поэтому вводим величину 10 000 со знаком минус (в формуле ячейка C2 вводится со знаком минус). У нас нет пополнений счета, но, если бы они были, значения ПЛТ (ячейка C4) также были бы со знаком минус.

А теперь разберем подробнее, как эту самую функцию БС (FV) из Excel извлечь.

Для того чтобы использовать встроенные функции в Excel, нам нужно ввести в ячейку C3 знак =, а затем нажать значок **fx**. Появится окно «Вставка функции», где в окошко «Найти» вводим нашу искомую функцию **БС (FV)**, а затем нажимаем «Найти» (ил. 4).

Щелкнув два раза левой кнопкой мыши на найденную функцию, получим окно «Аргументы функции» (ил. 5), которое необходимо заполнить.

Ставка — это наши 6% годовых. Лучше не вбивать значения руками, а сделать ссылку на ячейку со значением (C5). В этом случае не придется каждый раз вводить функцию заново, если какое-то из значений изменится.

Кпер — это количество периодов. Напомню, Excel считает периодами. Ни годами, неделями или днями, а именно периодами. Это очень важно! Количество периодов равно количеству выплат со счета или количеству внесенных на счет.




Иллюстрация 4

Это имеет значение, если период меньше года, в противном случае периодом считается один год. Например, у нас нет пополнения счета или изъятия денег с него, значит, периодом будет год. Количество периодов в нашем примере равно трем (или ячейка C7*C8).

Плат — это сумма пополнения или изъятия денег. Если мы счет пополняем, «выпускаем деньги из рук», то это значение будет со знаком минус. Если мы деньги со счета забираем, то должны вводить это значение со знаком плюс. В нашем примере нет ни пополнений, ни изъятий — делаем ссылку на ячейку C4 со знаком минус. Даже если сейчас сумма пополнений равна нулю, есть смысл вставить в формулу именно ссылку



Аргументы функции ? X

БС

Ставка	<input type="text"/>	↑	=	число
Кпер	<input type="text"/>	↑	=	число
Плт	<input type="text"/>	↑	=	число
Пс	<input type="text"/>	↑	=	число
Тип	<input type="text"/>	↑	=	число

=

Возвращает будущую стоимость инвестиции на основе периодических постоянных (равных по величине сумм) платежей и постоянной процентной ставки.

Ставка процентная ставка за период. Например, при годовой процентной ставке в 6% для квартальной ставки используйте значение 6%/4.

Значение:

[Справка по этой функции](#)

Иллюстрация 5

на ячейку, а не абсолютное значение «0». Вдруг ваши планы меняются, а в случае использования ссылки формулу менять не придется.

ПС — приведенная стоимость. Это начальная сумма. Она может быть, а может отсутствовать, поэтому это необязательный аргумент. Если мы ничего не поставим, функция приравняет это значение к нулю. Если ПС все же есть, как в нашем случае, то помним, «деньги из рук» — знак минус (ячейка C2).

Тип — еще один необязательный аргумент, который учитывает, когда происходит начисление процентов — в начале периода или в конце. Если начисление в начале периода, то ставим «1», если в конце, то «0». В том случае, если мы ничего не ставим, функция автоматически ставит «0».

Заполняем все аргументы и нажимаем ОК. Пока все просто, верно?
А сейчас задача посложнее.

10 000 рублей помещены в банк на депозит под 6% годовых на 3 года.
Каждый месяц доносим по 500 рублей. Сложные проценты начисляются ежемесячно. Как найти величину депозита в конце срока?

Начальная сумма (ПС) по-прежнему = 10 000 рублей

Платеж (Плт) = 500 рублей

Ставка (Ставка) = 6%

Количество лет (n) = 3

Количество пополнений в год (m) = 12

Не буду вас загружать, выкладывая сложные формулы. Просто посмотрим, как это считает EXCEL.

В этом примере начисления происходят не раз в год, а раз в месяц. Поэтому вместо годовой ставки мы должны использовать ставку периода начисления, в нашем случае ставку за месяц. **Напоминаю, Excel считает периодами!**

Как ее найти? Если годовая ставка равна 6%, то мы применим формулу $СТАВКА/m = 6\%/12 = 0,5\%$.

Кроме того, в этой задаче у нас появился регулярный платеж Плт = 500 рублей. Помним, эти деньги мы «выпускаем из рук», а значит в таблице они будут со знаком минус.

И еще одно замечание. При вводе значения в формулу, вы, скорее всего, обратили внимание, что в первой задаче КПЕР (количество периодов) у нас было равно n. В этой задаче начисления происходят 12 раз в год в течение трех лет. Поэтому $КПЕР = m \cdot n = 36!$

Обратите внимание, как мы вводим ставку и Кпер.

Нажимаем ОК и вуаля.

The image shows an Excel spreadsheet with the following data:

	A	B	C
1			
2		Приведенная стоимость (ПС)	10 000.00 Р
3		Будущая стоимость (БС)	=БС(С5/С8;С8*С7;-С4;-С2;0)
4		Платежи (Плт)	500.00 Р
5		Ставка (СТАВКА)	6.00%
6		Количество периодов (КПЕР)	36
7		Количество лет (n)	3
8		Количество начислений в году (m)	12

The 'Arguments of the Function' dialog box for the BS function is open, showing the following arguments:

- Ставка (Rate): C5/C8 = 0.005
- Кпер (Nper): C8*C7 = 36
- Плт (Pmt): -C4 = -500
- Пс (PV): -C2 = -10000
- Тип (Type): 0 = 0

The result of the function is 31634.85773. The dialog box also includes a description: 'Возвращает будущую стоимость инвестиции на основе периодических постоянных (равных по величине сумм) платежей и постоянной процентной ставки.' and a note about the 'Type' argument: 'Тип значение 0 или 1, обозначающее, должна ли выплата производиться в начале периода (1) или же в его конце (0 или отсутствие значения).'

Иллюстрация 6



Стоимость депозита составит 31 634,86 р. Если функция занесена правильно, то мы можем менять любой из параметров нашей таблицы, не меняя формулу.

Поскольку в окне «Аргументы функции» мы используем ссылки на ячейки, а не абсолютные значения, то можем просто менять любое значение, не исправляя формулу, и получать новое значение.

Например, изменим начальную сумму на 20 000 рублей.

	A	B	C	D
1				
2		Приведенная стоимость (ПС)	20 000.00 Р	
3		Будущая стоимость (БС)	43 601.66 Р	=БС(С5/С8;С8*С7;-С4;-С2;0)
4		Платежи (ПЛТ)	500.00 Р	
5		Ставка (СТАВКА)	6.00%	
6		Количество периодов (КПЕР)	36	
7		Количество лет (п)	3	
8		Количество начислений в году (m)	12	
9				

Иллюстрация 7

Или изменим ставку на 8,5%.

	A	B	C	D
1				
2		Приведенная стоимость (ПС)	20 000.00 Р	
3		Будущая стоимость (БС)	46 207.37 Р	=БС(С5/С8;С8*С7;-С4;-С2;0)
4		Платежи (ПЛТ)	500.00 Р	
5		Ставка (СТАВКА)	8.50%	
6		Количество периодов (КПЕР)	36	
7		Количество лет (п)	3	
8		Количество начислений в году (m)	12	
9				

Иллюстрация 8

Как видите, формула остается неизменной и позволяет варьировать параметры по нашему усмотрению.

С будущей стоимостью наших инвестиций, я надеюсь, мы разобрались. Но для закрепления давайте рассмотрим еще один пример.

Молодой человек решил откладывать «на старость» ежедневно по 50 рублей. Ставка 5,5%, срок инвестирования 40 лет. Первоначальных накоплений нет. Какова будет стоимость портфеля при выходе на пенсию?

У нас есть приведенная стоимость инвестиций (ПС), и она равна 0. Платеж (ПЛТ) равен 50 рублям. Ставка 5,5%. Количество лет — $n = 40$. Количество начислений в году — $m = 365$ (не будем учитывать високосные года). Вносим данные в таблицу для расчетов.

	A	B	C	D
1				
2		Приведенная стоимость (ПС)	0.00 Р	
3		Будущая стоимость (БС)	2 662 349.10 Р	=БС(C5/C8;C8*C7;-C4;-C2;0)
4		Платежи (ПЛТ)	50.00 Р	
5		Ставка (СТАВКА)	5.50%	
6		Количество периодов (КПЕР)	14600	
7		Количество лет (n)	40	
8		Количество начислений в году (m)	365	
9				

Иллюстрация 9

Стоимость портфеля составит 2 662 349,1 рубля. Нам не нужно каждый раз строить новые таблицы для вычислений. Мы можем пользоваться одной, заменяя исходные данные.

Итак, функция БС (FV) позволяет рассчитать будущую стоимость наших инвестиций при прочих заданных параметрах. Да, у нее есть ограни-



чения, она работает только для аннуитетных платежей. Но нужно помнить, что большая часть платежей, о которых идет речь, таковыми и являются.

Функция ПС (PV) (Приведенная стоимость)

Поговорим еще об одной функции Excel, которая поможет нам в определении настоящей стоимости будущих финансовых потоков. Рассмотрим на примерах, как ее можно использовать.

Интуитивно понятно, что деньги прямо сейчас и через 5 лет имеют разную стоимость. Мало кто согласится обменять тысячу рублей прямо сейчас на тысячу через 10 лет. Причин этому много: и инфляция, которая незаметно подтачивает покупательную способность денег, и альтернативные варианты размещения денег на этот срок. В экономике у этого явления есть собственное имя — «Теория временной стоимости денег».

Зачем менять тысячу сегодня на тысячу же через 5 лет, когда можно положить ее в банк и через те же пять лет получить, например, 1338,23 рубля?

А что это значит? Все просто: тысяча, которую мы собираемся получить через 5 лет, сегодня тоже не тысяча, а несколько меньше. Насколько меньше? Это мы и определим с помощью функции **ПС** (в английском варианте функция называется **PV**, или **present value**).

	A	B	C	D
1				
2		Приведенная стоимость (ПС)	1 000.00 Р	
3		Будущая стоимость (БС)	1 338.23 Р	=БС(C5/C8;C8*C7;-C4;-C2;0)
4		Платежи (ПЛТ)	0.00 Р	
5		Ставка (СТАВКА)	6.00%	
6		Количество периодов (КПЕР)	5	=C7*C8
7		Количество лет (n)	5	
8		Количество начислений в году (m)	1	
9				

Иллюстрация 10

Итак, сколько сегодня стоит тысяча рублей, которую мы получим через 5 лет?

Может показаться, что это не прикладной вопрос, однако в инвестиционном анализе расчет сегодняшней стоимости будущих денежных потоков используется существенно чаще, чем расчет будущей стоимости (БС (FV)).

Давайте рассмотрим очередной пример.

В течение последующих пяти лет вам нужно оплачивать образование ребенка, по 250 000 рублей в год. Какую сумму вам сейчас нужно разместить на депозит, чтобы ее хватило на все пять платежей? Ставка на пять лет 7%.

Алгоритм остается неизменным, ставим в ячейку, в которой хотим получить значение С2, знак = и переходим в окно «Вставка функции». Ищем функцию ПС, щелкаем по ней дважды левой кнопкой мыши и заполняем окно «Аргументы функции».

The image shows an Excel spreadsheet with the following data:

	A	B	C
1			
2		Приведенная стоимость (ПС)	=ПС(С5/С8;С7*С8;С4;С3;0)
3		Будущая стоимость (БС)	0,00 Р
4		Платежи (ПЛТ)	250 000,00 Р
5		Ставка (СТАВКА)	7,00%
6		Количество периодов (КПЕР)	5
7		Количество лет (n)	5
8		Количество начислений в году (m)	1
9			
10			
11			
12			

Overlaid on the spreadsheet is the 'Аргументы функции' (Function Arguments) dialog box for the 'ПС' (PV) function. The arguments are:

- Ставка (Rate): C5/C8 = 0.07
- Кпер (Nper): C7*C8 = 5
- Плт (Pmt): C4 = 250000
- Бс (Fv): C3 = 0
- Тип (Type): d = 0

The dialog also shows the resulting value: -1 025 049,36 Р. Below the dialog, the text 'Иллюстрация 11' is centered.

Иллюстрация 11

Обратите внимание, параметр m в нашем случае равен 1. Но я все равно ввожу полноценную формулу для того, чтобы в дальнейшем мы могли менять любые параметры задачи, не меняя саму формулу.

250 000 рублей мы забираем из портфеля, но они попадают «в руки», а «не из рук». Поэтому со знаком +.



Будущая стоимость в нашем случае равна 0, так как мы планируем израсходовать все средства на оплату обучения.

	A	B	C	D
1				
2		Приведенная стоимость (ПС)	-1 025 049.36 ₺	=ПС(C5/C8;C7*C8;C4;C3;0)
3		Будущая стоимость (БС)	0.00 ₺	
4		Платежи (ПЛТ)	250 000.00 ₺	
5		Ставка (СТАВКА)	7.00%	
6		Количество периодов (КПЕР)	5	=C7*C8
7		Количество лет (n)	5	
8		Количество начислений в году (m)	1	
9				

Иллюстрация 12

Итого под 7% на пять лет нужно разместить 1 025 049,36 рубля. Из них мы сможем ежегодно платить по 250 000 рублей за обучение ребенка. Давайте рассмотрим еще один пример применения функции ПС (PV).

Вам предложили вложиться в крутой бизнес. Сейчас вы должны заплатить 10 000 долларов, но затем в течение семи лет будете получать ежегодные выплаты в размере, указанном в таблице. Выгодно вам это или нет?

ГОД	ПЛАТЕЖ В КОНЦЕ ГОДА
1	\$ 2000,00
2	\$ 2000,00
3	\$ 2000,00
4	\$ 2000,00
5	\$ 2000,00
6	\$ 2000,00
7	\$ 2000,00

Иллюстрация 13

Для решения этой задачи нужно понять, какой сумме сегодня эквивалентен этот поток платежей.

Чтобы оценить, сколько этот поток платежей стоит сегодня, в данном случае мы вполне можем применить функцию ПС (PV). Платежи у нас аннуитетные, срок инвестирования n равен семи годам, платеж составляет 2000 долларов. И получаем мы его раз в год (то есть $m=1$).

У нас нет ставки, скажете вы, и будете правы. Ведь чтобы узнать приведенную стоимость, нам нужна ставка.

У этой задачи есть много решений. И даже зная пока всего две функции, мы можем попытаться найти ответ. Ведь вопрос стоит так: **выгодно или нет**.

Нам предлагают вложить 10 000 долларов для того, чтобы потом в течение 7 лет получать по 2000 долларов ежегодно. Выглядит заманчиво, но только в том случае, если у нас нет альтернативных вариантов вложить эти деньги. Представим, что такой вариант есть, и мы можем разместить наши деньги с приемлемым или схожим уровнем риска под 8% годовых на те же семь лет. В этом случае ставку можно принять равной 8% (но только если уровень риска схож).

Находим ПС (PV) нашего потока платежей при ставке 8%.

	A	B	C	D
1				
2		Приведенная стоимость (ПС)	-\$10 412.74	=ПС(C5/C8;C7*C8;C4;C3;0)
3		Будущая стоимость (БС)	\$0.00	
4		Платежи (ПЛТ)	\$2 000.00	
5		Ставка (СТАВКА)	8.00%	
6		Количество периодов (КПЕР)	7	=C7*C8
7		Количество лет (n)	7	
8		Количество начислений в году (m)	1	
9				

Иллюстрация 14

Как объяснить полученный результат (ил. 14) простыми словами? Для того чтобы мы, получая по 2000 долларов в год, зарабатывали в среднем



по 8% годовых в течение 7 лет, изначально нужно инвестировать 10 412,74 доллара. Нам же предлагают купить аналогичный поток платежей дешевле, всего за 10 000 долларов. Значит, на этом проекте мы сможем заработать больше, чем 8% годовых, и надо соглашаться.

А вот какую доходность мы получим, инвестируя 10 000 долларов, можно посчитать, используя функцию СТАВКА (RATE).

Функция СТАВКА (RATE)

Ставка — ключевой параметр любых финансовых вычислений. Если мы строим прогнозы относительно стоимости своих накоплений в будущем, то нужно выбрать ставку, на которую мы и будем ориентироваться. Желаемая ставка — это ставка, при которой при прочих заданных параметрах мы сможем достичь своей финансовой цели за определенный промежуток времени и с приемлемым для нас уровнем риска.

Ставка помогает нам не только строить прогнозы, но и принимать взвешенные инвестиционные решения.

Давайте не будем усложнять и сразу перейдем к примеру. Попробуем решить предыдущую задачку про 7 лет и 10 000 долларов.

Что мы имеем:

ПС = 10 000\$

БС = 0 \$

ПЛАТ = 2000\$

$n = 7$

$m = 1$

	A	B	C	D	E	F	G
1							
2		Приведенная стоимость (ПС)	\$10 000.00	<div style="border: 1px solid gray; padding: 5px;"> <p>Аргументы функции</p> <p>СТАВКА</p> <p>Кпер C7:C8 = 7</p> <p>Плт C4 = 2000</p> <p>Пс -C2 = -10000</p> <p>Бс C3 = 0</p> <p>Тип 0 = 0</p> <p>= 0.091961367</p> <p>Возвращает процентную ставку по аннуитету за один период. Например, при годовой процентной ставке в 6% для квартальной ставки используется значение 6%/4.</p> <p>Тип логическое значение (0 или 1), обозначающее, должна ли выплата производиться в конце периода (0) или отсутствие значения) или же в его начале (1).</p> <p>Значение: 9.20%</p> <p>Справка по этой функции</p> <p style="text-align: right;">OK Отмена</p> </div>			
3		Будущая стоимость (БС)	\$0.00				
4		Платежи (ПЛАТ)	\$2 000.00				
5		=СТАВКА(C7*C8;C4;-C2;C3;0)					
6		Количество периодов (КПЕР)	7				
7		Количество лет (n)	7				
8		Количество начислений в году (m)	1				
9							
10							
11							

Иллюстрация 15

Приведенная стоимость равна 10 000 долларов, она со знаком минус, так как мы деньги отдаем. Количество периодов не забываем вводить ссылками на ячейки — так гораздо удобнее. Платежи в нашем случае (ячейка С4) со знаком плюс, поскольку мы эти деньги получаем.

Нажимаем ОК и получаем нашу доходность.

	A	B	C	D
1				
2		Приведенная стоимость (ПС)	\$10 000.00	
3		Будущая стоимость (БС)	\$0.00	
4		Платежи (ПЛТ)	\$2 000.00	
5		Ставка (СТАВКА)	9.20%	=СТАВКА(С7*С8;С4;-С2;С3;0)
6		Количество периодов (КПЕР)	7	=С7*С8
7		Количество лет (n)	7	
8		Количество начислений в году (m)	1	
9				

Иллюстрация 16

Доходность данной инвестиции равняется 9,2% годовых, что, несомненно, лучше альтернативного варианта с доходностью 8% годовых.

Это очень важная опция — возможность определять ставку. Она позволяет нам при прочих заданных условиях выявить **желаемую** ставку для наших инвестиций. А кроме того, во-первых, понять, насколько мы объективно оцениваем реальность и возможности рынка, а во-вторых, что очень важно, сформировать структуру нашего портфеля.

Давайте разберем на примере.

У нас есть накопленный 1 000 000 рублей. Через 5 лет мы хотели бы купить дом. Ориентировочная цена дома 8 000 000 рублей. Мы можем пополнять счет ежемесячно на 50 000 рублей. Под какую ставку нам следует инвестировать деньги?



	A	B	C	D	E	F	G	H	I	J
1										
2		Приведенная стоимость (ПС)	1 000 000.00 ₽							
3		Будущая стоимость (БС)	8 000 000.00 ₽							
4		Платежи (ПЛТ)	50 000.00 ₽							
5		Ставка (СТАВКА)	=СТАВКА(C7*C8;-C4;-C2;C3;0)							
6		Количество периодов (КПЕР)	60.00							
7		Количество лет (n)	5							
8		Количество начислений в году (m)	12							
9										
10										
11										
12										
13										
14										
15										

Аргументы функции

СТАВКА

Кпер C7*C8 = 60

Плт -C4 = -50000

Пс -C2 = -1000000

Бс C3 = 8000000

Тип 0 = 0

= 0.017309037

Возвращает процентную ставку по аннуитету за один период. Например, при годовой процентной ставке в 6% для квартальной ставки используется значение 6%/4.

Тип логическое значение (0 или 1), обозначающее, должна ли выплата производиться в конце периода (0 или отсутствие значения) или же в его начале (1).

Значение: 1.73%

Справка по этой функции

OK Отмена

Иллюстрация 17

Вводим все значения, не забываем минус у ПЛТ и ПС, нажимаем ОК и получаем значение 1,73% (ил. 18). Но радоваться рано: вспоминаем, что Excel считает периодами. Сколько периодов в году, столько раз он капитализирует проценты (если период равен году, то капитализация происходит раз в год). Если мы доносим или выводим средства чаще, чем раз в год, то количество периодов в году у нас равно количеству довнесений/изъятий. В нашем варианте мы доносим деньги 12 раз в год. Значит, в году у нас 12 периодов. **А сейчас самое важное — СТАВКУ Excel посчитал нам тоже за период, то есть за 1 месяц.**

	A	B	C	D
1				
2		Приведенная стоимость (ПС)	1 000 000.00 ₽	
3		Будущая стоимость (БС)	8 000 000.00 ₽	
4		Платежи (ПЛТ)	50 000.00 ₽	
5		Ставка (СТАВКА)	1.73%	=СТАВКА(C7*C8;-C4;-C2;C3;0)
6		Количество периодов (КПЕР)	60	=C7*C8
7		Количество лет (n)	5	
8		Количество начислений в году (m)	12	
9				

Иллюстрация 18

И эту ставку нам еще предстоит привести к процентам годовых.

Как это сделать?

Простой путь — умножить 1,73% на 12.

Сложный путь — использовать формулу с учетом капитализации:

$$r_y = (1 + r_p)^m - 1,$$

где r_y — годовая доходность;
 r_p — доходность за период;
 m — количество периодов.

	A	B	C	D
1				
2		Приведенная стоимость (ПС)	1 000 000.00 Р	
3		Будущая стоимость (БС)	8 000 000.00 Р	
4		Платежи (ПЛТ)	50 000.00 Р	
5		Ставка (СТАВКА)	1.73%	=СТАВКА(С7*С8;-С4;-С2;С3;0)
6		Количество периодов (КПЕР)	60	=С7*С8
7		Количество лет (n)	5	
8		Количество начислений в году (m)	12	
9				
10		Ставка за период	1.73%	=С5
11		Количество периодов	12	=С8
12		Вариант 1	20.77%	=С10*С11
13		Вариант 2	22.87%	=(1+С10)^С11-1

Иллюстрация 19

Упс... Ставка уже не выглядит столь реалистично. А это значит, что риск не достигнуть поставленной финансовой цели очень и очень высок. Что делать? Пересмотреть какие-либо из параметров задачи: увеличить срок, увеличить ежемесячное пополнение, увеличить начальную сумму, умерить аппетит и выбрать дом попроще.

Самое простое — увеличить срок. Это действие от нас никаких дополнительных материальных затрат не потребует. Разумеется, нам придется

	A	B	C	D
1				
2		Приведенная стоимость (ПС)	1 000 000.00 Р	
3		Будущая стоимость (БС)	8 000 000.00 Р	
4		Платежи (ПЛТ)	50 000.00 Р	
5		Ставка (СТАВКА)	0.82%	=СТАВКА(С7*С8;-С4;-С2;С3;0)
6		Количество периодов (КПЕР)	84	=С7*С8
7		Количество лет (n)	7	
8		Количество начислений в году (m)	12	
9				
10		Ставка за период	0.82%	=С5
11		Количество периодов	12	=С8
12		Вариант 1	9.85%	=С10*С11
13		Вариант 2	10.31%	=(1+С10)^С11-1

Иллюстрация 20



чуть дольше пополнять счет на 50 000 рублей ежемесячно, но эта сумма вписывается в рамки нашей стратегии. Изменим инвестиционный горизонт на 2 года и посмотрим результат (см. ил. 20).

Этот вариант уже существенно ближе к реальности, и с ним можно работать.

Наша задача — получить адекватную желаемую ставку. Конечно, сразу возникает вопрос, а какую ставку можно считать адекватной? Мы не можем предсказывать будущее, но мы можем ориентироваться на прошлое. Если российский рынок ценных бумаг на протяжении последних 15 лет показывал среднюю доходность в районе 10% годовых, то СТАВКА 10% и меньше вполне могут считаться адекватными в отличие от 30 или 40% годовых, на которые рассчитывают многие инвесторы, пришедшие на рынок.

Можно попробовать изменить ежемесячное пополнение (это хороший вариант, но он, к сожалению, не всегда возможен).

Итак, увеличим сумму пополнений до 60 000 рублей.

	A	B	C	D
1				
2		Приведенная стоимость (ПС)	1 000 000.00 ₺	
3		Будущая стоимость (БС)	8 000 000.00 ₺	
4		Платежи (ПЛТ)	60 000.00 ₺	
5		Ставка (СТАВКА)	1.43%	=СТАВКА(C7*C8;-C4;-C2;C3;0)
6		Количество периодов (КПЕР)	60	=C7*C8
7		Количество лет (n)	5	
8		Количество начислений в году (m)	12	
9				
10		Ставка за период	1.43%	=C5
11		Количество периодов	12	=C8
12		Вариант 1	17.22%	=C10*C11
13		Вариант 2	18.65%	=(1+C10)^C11-1

Иллюстрация 21

Этот вариант хуже, он не позволяет нам добиться приемлемого значения СТАВКИ.

Можно решать эту задачу методом перебора, а можно использовать возможности Excel и построить таблицы чувствительности.

Разберем это на нашем простом примере.

Таблицы чувствительности

Таблицы чувствительности позволяют не заниматься перебором параметров, а посмотреть, так сказать, на картину в целом. Давайте рассмотрим всю ту же задачу. Для этого выберем параметр, который будем менять. Пусть это будет **ежемесячное пополнение**.

Шаг 1

В ячейке В16, используя ссылку на ячейку С13, размещаем значение ставки.

Шаг 2

В ячейках С15:F15 размещаем варианты ежемесячного пополнения портфеля (например, 50 000, 55 000, 60 000 и т. д.)

	A	B	C	D	E	F
2		Приведенная стоимость (ПС)	1 000 000.00 ₽			
3		Будущая стоимость (БС)	8 000 000.00 ₽			
4		Платежи (ПЛТ)	60 000.00 ₽			
5		Ставка (СТАВКА)	1.43%			
6		Количество периодов (КПЕР)	60			
7		Количество лет (n)	5			
8		Количество начислений в году (m)	12			
9						
10		Ставка за период	1.43%	=C5		
11		Количество периодов	12	=C8		
12		Вариант 1	17.22%	=C10*C11		
13		Вариант 2	18.65%	=(1+C10)^C11-1		
14						
15			50000	55000	60000	65000
16		18.65%				

Таблица данных

Подставлять значения по столбцам в:

Подставлять значения по строкам в:

OK Отмена

Иллюстрация 22

Шаг 3

Выделяем ячейки В15:F16 и идем во вкладку «Данные» — «Анализ что, если» — «Таблица данных».

В Таблицу данных подставляем ссылку на ячейку С4 в строку «Подставлять значения по столбцам в». В столбцах у нас ежемесячные пополнения, значит, именно на эту ячейку мы и делаем ссылку.

Нажимаем ОК и получаем спектр значений СТАВКИ, соответствующий указанным нами ежемесячным платежам.



	A	B	C	D	E	F
1						
2		Приведенная стоимость (ПС)	1 000 000.00 Р			
3		Будущая стоимость (БС)	8 000 000.00 Р			
4		Платежи (ПЛТ)	60 000.00 Р			
5		Ставка (СТАВКА)	1.43%	=СТАВКА(C7*С8;-С4;-С2;С3;0)		
6		Количество периодов (КПЕР)	60	=С7*С8		
7		Количество лет (п)	5			
8		Количество начислений в году (m)	12			
9						
10		Ставка за период	1.43%	=С5		
11		Количество периодов	12	=С8		
12		Вариант 1	17.22%	=С10*С11		
13		Вариант 2	18.65%	=(1+С10)^С11-1		
14						
15			50000	55000	60000	65000
16			18.65%	22.87%	20.71%	18.65% 16.67%

Иллюстрация 23

Подставляем в графу «Подставить значения по столбцам» ссылку на ту ячейку, аргумент в которой мы исследуем. В нашем случае это ежемесячное пополнение счета, ячейка С4.

Результат готов. Таблица чувствительности дает нам наглядное представление, как размер ежемесячного пополнения счета влияет на желаемую ставку, которая нам необходима для достижения нашей финансовой цели.

Результат не очень обнадеживает. Поэтому часто возникает необходимость провести такой анализ не по одному параметру, а по двум. Давайте это сделаем.

Алгоритм схожий, но нам нужно выбрать второй параметр, по которому мы будем проводить анализ. Вариантов не так много, рассмотрим срок до цели. Возьмем интервал изменений 0,5 года и построим новую таблицу чувствительности.

Выглядеть она будет как на ил. 24. Не забываем, что в ячейке В18 у нас не число, а ссылка на ячейку, которая считает (ячейка С13).

Выделяем всю таблицу В18:F25 и идем по старому маршруту: «Данные» — «Анализ что, если» — «Таблица данных».

По столбцам делаем ссылку на ячейку С4 (ежемесячное пополнение), а по строкам — на ячейку С7 (количество лет).

Нажимаем ОК и получаем целую таблицу значений для СТАВКИ при разных сроках инвестирования и разных суммах пополнения (ил. 25)

ГЛАВА 2

	A	B	C	D	E	F
1						
2		Приведенная стоимость (ПС)	1 000 000.00 Р			
3		Будущая стоимость (БС)	8 000 000.00 Р			
4		Платежи (ПЛТ)	60 000.00 Р			
5		Ставка (СТАВКА)	1.43%			
6		Количество периодов (КПЕР)	60			
7		Количество лет (n)	5			
8		Количество начислений в году (m)	12			
9						
10		Ставка за период	1.43%	=C5		
11		Количество периодов	12	=C8		
12		Вариант 1	17.22%	=C10*C11		
13		Вариант 2	18.65%	=(1+C10)^C11-1		
14						
15			50000	55000	60000	65000
16		18.65%	22.87%	20.71%	18.65%	16.67%
17						
18		18.65%	50000	55000	60000	65000
19		5				
20		5.5				
21		6				
22		6.5				
23		7				
24		7.5				
25		8				

Таблица данных ? X

Подставлять значения по столбцам в: \$C\$4 ↑

Подставлять значения по строкам в: \$C\$7 ↑

OK Отмена

Иллюстрация 24

	A	B	C	D	E	F
1						
2		Приведенная стоимость (ПС)	1 000 000.00 Р			
3		Будущая стоимость (БС)	8 000 000.00 Р			
4		Платежи (ПЛТ)	60 000.00 Р			
5		Ставка (СТАВКА)	1.43%	=СТАВКА(C7^C8)-C4)-C2)/3n)		
6		Количество периодов (КПЕР)	60	=C7^C8		
7		Количество лет (n)	5			
8		Количество начислений в году (m)	12			
9						
10		Ставка за период	1.43%	=C5		
11		Количество периодов	12	=C8		
12		Вариант 1	17.22%	=C10^C11		
13		Вариант 2	18.65%	=(1+C10)^C11-1		
14						
15			50000	55000	60000	65000
16		18.65%	22.87%	20.71%	18.65%	16.67%
17						
18		18.65%	50000	55000	60000	65000
19		5	22.87%	20.71%	18.65%	16.67%
20		5.5	18.65%	16.62%	14.68%	12.83%
21		6	15.29%	13.37%	11.54%	9.78%
22		6.5	12.56%	10.73%	8.99%	7.32%
23		7	10.31%	8.56%	6.90%	5.30%
24		7.5	8.42%	6.74%	5.15%	3.62%
25		8	6.82%	5.21%	3.67%	2.20%
26						

Иллюстрация 25



Сразу видно, где ставка становится «приличной», то есть рабочей.

Эта нехитрая процедура сильно упрощает процесс принятия инвестиционных решений и делает финансовый анализ куда более приятным занятием, чем при расчетах без применения Excel.

Ну и для закрепления материала вернемся к задачке про Петю и его накопления на квартиру.

У Пети есть 1 000 000 рублей. Через три года ему нужно 3 000 000 рублей для того, чтобы оплатить первоначальный взнос за квартиру. Как ему построить свой финансовый план?

У нас есть начальная и конечные суммы, а также период инвестирования. Нет ставки и нет суммы ежемесячных пополнений счета.

Вариантов решения, как всегда, много. Для начала можно попробовать вычислить ставку для варианта, когда довнесений на счет просто нет.

	A	B	C	D
1				
2		Приведенная стоимость (ПС)	1 000 000.00 ₽	
3		Будущая стоимость (БС)	3 000 000.00 ₽	
4		Платежи (ПЛТ)	0.00 ₽	
5		Ставка (СТАВКА)	44.22%	=СТАВКА(C7*C8;-C4;-C2;C3;0)
6		Количество периодов (КПЕР)	3	=C7*C8
7		Количество лет (n)	3	
8		Количество начислений в году (m)	1	
9				

Иллюстрация 26

Полученная ставка оптимизма не внушает, но теперь мы точно знаем, что без пополнения счета никак не обойтись.

Можно попробовать построить таблицу чувствительности для аргумента СТАВКА, меняя размер ежемесячного пополнения счета (ил. 27).

Заметьте, если мы вводим ежемесячное пополнение счета, то количество начислений в год в нашем примере изменится с 1 на 12 (ячейка С8). Соответственно, и количество периодов изменится с 3 до 36 (ячейка С6).

	A	B	C	D	E	F
1						
2		Приведенная стоимость (ПС)	1 000 000.00 Р			
3		Будущая стоимость (БС)	3 000 000.00 Р			
4		Платежи (ПЛТ)	0.00 Р			
5		Ставка (СТАВКА)	37.18%	=СТАВКА(C7*C8;-C4;-C2;C3;0)*C8		
6		Количество периодов (КПЕР)	36	=C7*C8		
7		Количество лет (n)	3			
8		Количество начислений в году (m)	12			
9						
10			10000	20000	30000	40000
11		37.18%				
12						

Таблица данных ? X

Подставлять значения по столбцам в: ↑

Подставлять значения по строкам в: ↑

OK Отмена

Иллюстрация 27

Изменится и СТАВКА, так как Excel считает ставку за период. А период у нас изменился. Был 1 год, а стал 1 месяц. Поэтому в ячейке C5 СТАВКА я сразу умножу ставку на количество начислений в году (ячейка C8).

Выделяем диапазон B10:F11. Данные — Анализ «что, если» — Таблицы данных. В графу «Подставлять значения по столбцам в» вводим ячейку C4 (функция ПЛТ).

В результате получим таблицу чувствительности ставки к сумме ежемесячных пополнений, из которой можно сделать следующий вывод: для получения приемлемой ставки ежемесячные пополнения счета должны быть не меньше 40 000 рублей. Только в этом случае финансовый план по достижении поставленной цели выглядит реалистично.

	A	B	C	D	E	F
1						
2		Приведенная стоимость (ПС)	1 000 000.00 Р			
3		Будущая стоимость (БС)	3 000 000.00 Р			
4		Платежи (ПЛТ)	0.00 Р			
5		Ставка (СТАВКА)	37.18%	=СТАВКА(C7*C8;-C4;-C2;C3;0)*C8		
6		Количество периодов (КПЕР)	36	=C7*C8		
7		Количество лет (n)	3			
8		Количество начислений в году (m)	12			
9						
10			10000	20000	30000	40000
11		37.18%	29.93%	22.94%	16.21%	9.70%
12						

Иллюстрация 28